


# SMALLTALK-80

THE LANGUAGE AND ITS IMPLEMENTATION



Adele Goldberg and David Robson

class name FinancialHistory  
superclass Object  
instance variable names cashOnHand  
incomes  
expenditures

message category

class methods

instance creation

statements

message pattern

**initialBalance: amount**

↑super new setInitialBalance: amount

**new**

↑super new setInitialBalance: 0

instance methods

transaction recording

**receive: amount from: source**

incomes at: source

put: (self totalReceivedFrom: source) + amount.

cashOnHand ← cashOnHand + amount

**spend: amount for: reason**

expenditures at: reason

put: (self totalSpentFor: reason) + amount.

cashOnHand ← cashOnHand - amount

inquiries

**cashOnHand**

↑cashOnHand

**totalReceivedFrom: source**

(incomes includesKey: source)

ifTrue: [↑incomes at: source]

ifFalse: [↑0]

**totalSpentFor: reason**

(expenditures includesKey: reason)

ifTrue: [↑expenditures at: reason]

ifFalse: [↑0]

private

**setInitialBalance: amount**

cashOnHand ← amount.

incomes ← Dictionary new.

expenditures ← Dictionary new

message category

message pattern

functional specification

FinancialHistory class protocol

class initialization

initialBalance: amount

Begin a financial history with amount as the amount of money on hand.

new

Begin a financial history with 0 as the amount of money on hand.

FinancialHistory instance protocol

transaction recording

receive: amount from: source

Remember that an amount of money, amount, has been received from source.

spend: amount for: reason

Remember that an amount of money, amount, has been spent for reason.

inquiries

cashOnHand

Answer the total amount of money currently on hand.

totalReceivedFrom: source

Answer the total amount received from source, so far.

totalSpentFor: reason

Answer the total amount spent for reason, so far.